Тема 2. Начальный загрузчик


Основным назначением загрузочного сектора дискет и жёстких дисков является загрузка начального кода ОС с диска в память и передача ему управления. Помимо кода, считывающего в память начальный код ОС, загрузочный сектор также содержит некоторую важную информацию о диске – тип ОС, тип файловой системы, количество секторов и т. д. Далее будет рассмотрена структура загрузочного сектора, используемая совместно с файловыми системами FAT. Файловая система NTFS использует другую структуру загрузочного сектора.
Структура загрузочного сектора дискет и жёстких дисков

В таблице 1 представлена структура загрузочного сектора дискет и жёстких дисков с указанием смещений и размеров полей структуры.

Таблица 1 – структура загрузочного сектора дискет и жёстких дисков.

	Смещение поля 
(в байтах)
	Размер поля 
(в байтах)
	Пример 
значения 
в поле
	Назначение поля

	0x0000
	3
	EB 3C 90
	Команда перехода (JMP)

	0x0003
	8
	MSDOS5.0
	Тип ОС

	0x000B
	25
	
	Блок параметров BIOS

	0x0024
	26
	
	Расширенный блок параметров BIOS

	0x003E
	448
	
	Загрузочный код

	0x01FE
	2
	0x55AA
	Маркер конца сектора



Команда перехода практически всегда одинакова и передаёт управление на смещение 0x003E, где начинается загрузочный код. Выполнение команды перехода необходимо для того, чтобы «обойти» структуры данных, размещаемые перед загрузочным кодом. 
Тип ОС – это текстовая строка размером 8 байт, идентифицирующая ту ОС, которая установлена на дискете или на разделе жёсткого диска. Эта строка для MS-DOS версии 5.0 и выше всегда равна ‘MSDOS5.0’ для ОС Windows 95 эта строка равна ‘MSWIN4.0’, а для Windows 98 – ‘MSWIN4.1’.
Блоки параметров BIOS содержат важную информацию о структуре диска – количество байт в секторе, количество секторов в кластере, количество копий FAT и т. д. Структура блока параметров BIOS приведена в таблице 2, а структура расширенного блока параметров BIOS – в таблице 3.

Загрузочный код выполняет операции по считыванию начального кода ОС в память и передачу ему управления. Для считывания данных с диска в загрузочном коде используются функции прерывания BIOS Int 13h (работа с диском на низком уровне). Поскольку для загрузочного кода отводится только 448 байт, начальный код ОС должен располагаться в последовательных секторах для сокращения числа команд, используемых для получения доступа к этому коду.


Маркер конца сектора служит для локализации конца сектора. Маркер не используется загрузочным кодом, а обычно используется различными диагностическими дисковыми утилитами.

Таблица 2 – структура блока параметров BIOS.

	Смещение поля 
(в байтах)*
	Размер поля 
(в байтах)
	Назначение поля

	0x0B
	2
	Количество байт в секторе, т. е. размер физического сектора. Для большинства используемых дисков значение этого поля равно 512.

	0x0D
	1
	Число секторов в кластере. Первоначальный размер кластера зависит от размера тома и используемой файловой системы.

	0x0E
	2
	Количество зарезервированных секторов, т. е. количество секторов от загрузочного сектора до начала первой таблицы распределения файлов, включая сам загрузочный сектор. Минимальное значение этого поля: 1. Если значение в поле больше 1, это означает, что загрузочный код не умещается целиком в загрузочном секторе и занимает один или несколько следующих секторов.

	0x10
	1
	Количество таблиц распределения файлов (File Allocation Tables – FATs), т. е. количество копий FAT на данном томе. Обычно значение этого поля равно 2.

	0x11
	2
	Максимальное количество элементов корневого каталога, которое может храниться в корневой папке тома. Один элемент всегда используется для метки тома. Файлы с длинными именами используют несколько элементов оглавления. Обычно максимально количество файлов в корневом каталоге равно 511, т. е. значение данного поля обычно равно 512 (+1 элемент на метку тома). Однако, если в корневом каталоге используются длинные имена файлов, максимальное количество файлов будет меньше 511.

	0x13
	2
	Количество секторов в томе. Данное поле используется, если число секторов в томе не более 65535. Если же число секторов в томе больше 65535, то значение данного поля равно 0, и используется поле со смещением 0x20. 

	0x15
	1
	Тип носителя. Значение этого поля характеризует используемый тип носителя. Для жёстких дисков значение данного поля равно 0xF8.

	0x16
	2
	Количество секторов в таблице распределения файлов, т. е. количество секторов, занимаемое каждой копией таблицы распределения файлов тома. Используя информацию из этого поля, а также зная количество копий FAT и число зарезервированных секторов, можно вычислить номер сектора, где начинается корневой каталог. Подобные вычисления используются загрузочным кодом для получения доступа к файлу, хранящему начальный код загрузки ОС.

	0x18
	2
	Количество секторов на дорожке. Значение в это поле заносится при форматировании диска.

	0x1A
	2
	Количество головок (сторон) диска. Значение в это поле заносится при форматировании диска.

	0x1C
	4
	Количество скрытых секторов. Значение этого поля используется только для жёстких дисков. Значение поля характеризует смещение (в секторах) от начала диска до раздела, в котором размещается данный загрузочный сектор.

	0x20
	4
	Данное поле используется, если количество секторов на томе превышает 65535. Если же количество секторов меньше или равно 65535, используется поле со смещением 0x13.


* - здесь и далее смещения приводятся относительно начала загрузочного сектора.

Таблица 3 – структура расширенного блока параметров BIOS.
	Смещение поля 
(в байтах)
	Размер поля 
(в байтах)
	Назначение поля

	0x24
	1
	Физический номер диска. Этот параметр относится к физическому номеру диска BIOS. Гибкие диски нумеруются, начиная с 0x00 (диск A). Жёсткие диски нумеруются, начиная с 0x80 (диск C). Значение этого поля используется при вызове загрузочным кодом функций прерывания BIOS Int 13h.

	0x25
	1
	Текущий номер головки. Это поле не используется для файловых систем FAT. Windows NT использует данной поле для хранения двух флагов:

1. 0-й бит используется для определения необходимости запуска программы Chkdsk для данного диска при запуске Windows;

2. установленный 1-й бит говорит о необходимости запуска теста поверхности данного диска; данный бит имеет значение только если бит 0 установлен в 1.

	0x26
	1
	Сигнатура. Используется для обозначения корректного раздела или дискеты. Значение этого поля должно быть 0x28 или 0x29.

	0x27
	4
	Серийный номер тома. Уникальное число, которое записывается в данное поле при форматировании диска.

	0x2B
	11
	Метка тома. Раньше это поле использовалось для хранения метки тома, но в настоящее время для этой цели в корневом каталоге используется специальный файл с атрибутом ‘метка тома’.

	0x36
	8
	Идентификатор файловой системы. Служит для определения файловой системы, установленной на данном томе. Значение этого поля может быть равно ‘FAT12’, ‘FAT16’ или ‘FAT32’.



В таблице 4 представлены коды значения поля типа носителя (блок параметров BIOS) для различных типов дисководов.

Таблица 4 – коды значения поля типа носителя.

	Значение
	Ёмкость
	Размер и тип носителя

	0xF0
	2,88 Мб
	3,5”, 2 стороны, 36 секторов на дорожку

	0xF0
	1,44 Мб
	3,5”, 2 стороны, 18 секторов на дорожку

	0xF9
	720 Кб
	3,5”, 2 стороны, 9 секторов на дорожку

	0xF9
	1,2 Мб
	5,25”, 2 стороны, 15 секторов на дорожку

	0xFD
	360 Кб
	5,25”, 2 стороны, 9 секторов на дорожку

	0xFF
	320 Кб
	5,25”, 2 стороны, 8 секторов на дорожку

	0xFC
	180 Кб
	5,25”, 1 сторона, 9 секторов на дорожку

	0xFE
	160 Кб
	5,25”, 1 сторона, 8 секторов на дорожку



Далее будет рассмотрен формат загрузочного сектора для разделов, использующих файловую систему NTFS. В таблице 5 представлена структура загрузочного сектора разделов с указанием смещений и размеров полей структуры.

Таблица 5 – структура загрузочного сектора разделов с файловой системой  NTFS.

	Смещение поля 
(в байтах)
	Размер поля 
(в байтах)
	Пример 
значения 
в поле
	Назначение поля

	0x0000
	3
	EB 5B 00
	Команда перехода (JMP)

	0x0003
	8
	NTFS
	Тип файловой системы

	0x000B
	25
	
	Блок параметров BIOS

	0x0024
	48
	
	Расширенный блок параметров BIOS

	0x003E
	426
	
	Загрузочный код

	0x01FE
	2
	0x55AA
	Маркер конца сектора



Из сравнения таблиц 1 и 5 видно, что размер загрузочного кода в загрузочном секторе NTFS меньше, чем у FAT. Однако, размер загрузочного кода Windows NT/2000 больше 426 байт и при форматировании диска под загрузочный код отводятся первые 16 секторов. В этом случае количество зарезервированных секторов (поле по смещению 0x0E в блоке параметров BIOS) будет равно 16. 
Дополнительные поля расширенного блока параметров BIOS необходимы для локализации одной из основных структур NTFS – Master File Table (MFT). MFT в отличие от FAT не располагается в фиксированных секторах, что обеспечивает возможность перемещения MFT в другое место диска в случае возникновения повреждённых секторов в области MFT.

Структура блока параметров BIOS для томов NTFS представлена в таблице 6, а структура расширенного блока параметров – в таблице 7.

Таблица 6 – структура блока параметров BIOS для томов NTFS.
	Смещение поля 
(в байтах)
	Размер поля 
(в байтах)
	Назначение поля

	0x0B
	2
	Количество байт в секторе.

	0x0D
	1
	Число секторов в кластере.

	0x0E
	2
	Количество зарезервированных секторов.

	0x10
	1
	Значение этого поля всегда равно 0.

	0x11
	2
	Значение этого поля всегда равно 0.

	0x13
	2
	Не используется NTFS. 

	0x15
	1
	Тип носителя.

	0x16
	2
	Значение этого поля всегда равно 0.

	0x18
	2
	Количество секторов на дорожке.

	0x1A
	2
	Количество головок (сторон) диска.

	0x1C
	4
	Количество скрытых секторов.

	0x20
	4
	Не используется NTFS.


Таблица 7 – структура расширенного блока параметров BIOS для томов NTFS.
	Смещение поля 
(в байтах)
	Размер поля 
(в байтах)
	Назначение поля

	0x24
	4
	Не используется NTFS.

	0x28
	8
	Общее количество секторов в разделе NTFS.

	0x30
	8
	Логический номер кластера для файла MFT$.

	0x38
	8
	Логический номер кластера для файла MFT$Mirr.

	0x40
	4
	Количество кластеров в сегменте записи файла.

	0x44
	4
	Количество кластеров в индексном блоке.

	0x48
	8
	Серийный номер тома.

	0x50
	4
	Контрольная сумма.



Более подробно файловая система NTFS будет рассматриваться в теме 9 («Операционная система Windows 2000 Professional»).
Структура главной загрузочной записи жёстких дисков


Жёсткие диски могут содержать несколько разделов, на каждом из которых может быть установлена ОС. В этом случае каждый раздел имеет свой собственный загрузочный сектор. Информация о разделах хранится в так называемой главной загрузочной записи (Master Boot Record – MBR) жёсткого диска. Помимо информации о разделах в  MBR также хранится загрузочный код, который определяет, какой из разделов является активным, считывает его загрузочный сектор в память и передаёт ему управление. Главная загрузочная запись имеет размер 512 байт и занимает сектор 1 дорожки 0 стороны 0 жёсткого диска. Загрузочные сектора разделов размещаются в самых первых секторах соответствующих разделов.


MBR является основной структурой данных жёсткого диска. Существуют вирусы, которые модифицируют MBR, делая невозможным загрузку ОС. Рекомендуется устанавливать в BIOS опцию проверки MBR на вирусы. Эта опция доступна практически во всех современных BIOS.


Структура главной загрузочной записи жёсткого диска представлена в таблице 8.

Таблица 8 – структура главной загрузочной записи жёсткого диска.

	Смещение поля 
(в байтах)
	Размер поля 
(в байтах)
	Назначение поля

	0x0000
	440
	Загрузочный код. Данный код запускается BIOS при загрузке с жёсткого диска. Если на жёстком диске присутствуют разделы с установленной ОС, то загрузочный код определяет, какой из этих разделов является активным, считывает соответствующий загрузочный сектор и передаёт ему управление.

	0x01B8
	6
	Сигнатура диска. Это уникальное число, которое используется Windows NT для  хранения в реестре информации о диске (ветвь HKEY_LOCAL_MACHINE\SYSTEM\DISK).

	0x1BE
	64
	Таблица разделов. Содержит всю необходимую информацию о разделах жёсткого диска (начало, конец, размер, тип и т. д.)

	0x01FE
	2
	Маркер конца сектора


Типы разделов и таблица разделов


Всего жёсткий диск может содержать до 4 разделов. Поэтому таблица разделов хранит в себе 4 записи – по одной для каждого раздела. Разделы бывают основными и дополнительными. Дополнительный раздел может быть только один. Таким образом, на жёстком диске можно разместить максимум либо 3 основных и один дополнительный раздел, либо 4 основных раздела. 


Каждый раздел может иметь свою индивидуальную файловую систему. Например, можно выделить раздел с файловой системой NTFS для Windows NT и раздел с файловой системой FAT32 для Windows 98.

Основной раздел – это тот раздел, на котором установлены загрузочные файлы ОС. Сама ОС (её системные файлы) может располагаться на любом другом разделе. С основного раздела происходит загрузка ОС, поэтому основной раздел включает в себя загрузочный сектор.

Основной раздел может быть активным или неактивным, причём активным может быть только один основной раздел, если их несколько. Загрузочный код MBR выбирает для загрузки именно активный раздел и считывает его загрузочный сектор с последующей передачей ему управления.

Дополнительный раздел предназначен для организации логических дисков. С этого раздела не может осуществляться загрузка ОС. Сам по себе дополнительный раздел не добавляет логических дисков – они должны быть созданы в нём. Например, если жёсткий диск не является системным и содержит только один дополнительный раздел, то в этом дополнительном разделе можно разместить до 23 дисков (D, E, F, G, H, I, J, K, L, M, N, O, P, Q, R, S, T, U, V, W, X, Y, Z). Здесь предполагается, что буква C зарезервирована для системного диска.

Жёсткий диск не обязательно полностью заполнять разделами. Какая-то часть жёсткого диска может оставаться не разбитой на разделы. Это справедливо и для дополнительного раздела, часть которого также может оставаться не разбитой.

Структура таблицы разделов представлена в таблице 9.

Таблица 9 – структура таблицы разделов.

	Смещение поля в байтах
	Название поля

	0x01BE
	Раздел 1

	0x01CE
	Раздел 2

	0x01DE
	Раздел 3

	0x01EE
	Раздел 4



Общий размер таблицы разделов составляет 64 байта. Каждая запись в таблице занимает 16 байт и определяет характеристики одного из 4-х разделов жёсткого диска. Структура записей таблицы разделов представлена в таблице 10.

Таблица 10 – структура записей таблицы разделов.

	Смещение поля 
(в байтах)*
	Размер поля 
(в байтах)
	Назначение поля

	0x00
	1
	Индикатор загрузки. Определяет, является ли данный раздел активным (значение 0x80) или неактивным (значение 0x00).

	0x01
	1
	Начальная головка (сторона). Принимает значения от 0 до 255.

	0x02
	1
	Начальный сектор (биты 0-5). Это поле также содержит старшие 2 бита номера цилиндра (биты 6-7). Номер сектора может быть в пределах от 1 до 63.

	0x03
	1
	Начальный цилиндр. В данном поле содержатся младшие 8 бит номера цилиндра. Используя 2 бита из поля начального сектора, номер цилиндра занимает 10 бит и таки образом может быть в пределах от 0 до 1023.

	0x04
	1
	Системный идентификатор.

	0x05
	1
	Конечная головка.

	0x06
	1
	Конечный сектор (биты 0-5) и старшие 2 бита номера конечного цилиндра (биты 6-7).

	0x07
	1
	Младшие 8 бит номера конечного цилиндра.

	0x08
	4
	Относительный номер сектора. Показывает количество секторов от начала диска до начала раздела.

	0x12
	4
	Общее количество секторов в разделе.


* - смещения приведены относительно начала таблицы разделов.


Поле системного идентификатора определяет тип используемой файловой системы. Для Windows NT/2000 это поле также определяет то, какой драйвер файловой системы использовать при загрузке. Значение этого поля также может определять дополнительный раздел.


В таблице 11 представлены возможные значения поля системного идентификатора.

Таблица 11 – значения поля системного идентификатора.

	Значение поля
	Описание

	0x01
	Раздел или логический диск FAT12. Общее количество секторов в разделе не превышает 32679.

	0x04
	Раздел или логический диск FAT16. Общее количество секторов в разделе находится в пределах от 32680 до 65535.

	0x05
	Расширенный раздел.

	0x06
	Раздел или логический диск BIGDOS FAT.

	0x07
	Раздел или логический диск NTFS.



MS-DOS распознаёт только разделы с системным идентификатором 1, 4, 5 или 6.


Windows 9x также определяет и другие значения для поля системного идентификатора (см. табл. 12). Разделы с этими значениями не распознаются Windows NT, но распознаются Windows 2000/XP.
Таблица 12 – значения системного идентификатора, определяемые Windows 9x.

	Значение поля
	Описание

	0x0B
	Основной раздел FAT32 с использованием расширений прерывания Int 13h.

	0x0C
	Дополнительный раздел FAT32 с использованием расширений прерывания Int 13h.

	0x0E
	Основной раздел FAT16 с использованием расширений прерывания Int 13h.

	0x0F
	Дополнительный раздел FAT16 с использованием расширений прерывания Int 13h.



Поля начальной стороны, сектора и цилиндра используются MBR для считывания загрузочного сектора активного раздела в память.


Максимальный объём жёсткого диска, разделы которого могут быть описаны в таблице разделов зависит от максимального количества секторов, сторон и цилиндров. Поскольку современные жёсткие диски форматируются с использованием 512 байт на сектор, максимальный объём жёсткого диска для таблицы разделов можно вычислить по формуле:

MaxCapacity = sectors * cylinders * heads * 512 или
MaxCapacity = 63 * 1024 * 256 * 512 = 8455716864 байта (7,875 Гб).


Поскольку максимальный размер кластера для файловых систем FAT составляет 65536 байт, максимальный объём тома FAT составляет 4 Гб. Максимальный размер тома NTFS гораздо больше и составляет более 32 Гб.


Для жёстких дисков объёмом более 7,875 Гб доступ к разделом с использованием начальных сектора, стороны и цилиндра невозможен. Windows NT/2000 в этом случае используют относительный номер сектора, который показывает смещение (в секторах) раздела от начала жёсткого диска. Для таких жёстких дисков в BIOS Setup должен быть включён режим LBA, который вместо трёхмерной геометрии диска (цилиндр-головка-сектор) использует линейную адресацию по логическим номерам секторов.


Если требуется создание более 4-х логических дисков, то их можно сформировать в расширенном разделе. Основная таблица разделов указывает на первый сектор первого логического диска расширенного раздела, в которомхранится расширенная таблица разделов. Расширенная таблица разделов имеет такой же формат, что и таблица разделов в MBR. Записи расширенной таблицы разделов содержат следующую информацию:

1. Первая запись указывает на текущий логический диск (на себя) в расширенном разделе.

2. Вторая запись указывает на следующий логический диск в расширенном разделе.

3. Третья и четвёртая записи заполнены нулями и не ссылаются ни на один логический диск.

Первый сектор каждого логического диска в расширенном разделе имеет подобный формат и таким образом организуется цепочка логических дисков, каждый из которых указывает на последующий логический диск.

Использование полей относительного номера сектора и общего количества секторов в расширенной таблице разделов отличается от основной таблицы разделов.

Для записи 1 относительный номер сектора – это смещение (в секторах) от начала логического диска до загрузочного сектора раздела. Общее количество секторов показывает размер текущего раздела за вычетом загрузочного сектора.

Для записи 2 относительный номер сектора – это смещение от начала дополнительного раздела до первого сектора следующего логического диска, содержащего расширенную таблицу разделов. Общее количество секторов показывает объём следующего логического диска.

Ниже в качестве примера представлен шестнадцатеричный дамп главной загрузочной записи жёсткого диска объёмом 60 Гб с 2 основными и одним дополнительным разделом.

00000000: 33 C0 8E D0 BC 00 7C FB 50 07 50 1F FC BE 1B 7C | 3└О╨╝.|√P.P.№╛.|

00000010: BF 1B 06 50 57 B9 E5 01 F3 A4 CB BD BE 07 B1 04 | ┐..PW╣х.єд╦╜╛.▒.

00000020: 38 6E 00 7C 09 75 13 83 C5 10 E2 F4 CD 18 8B F5 | 8n.|.u.Г┼.тЇ═.Лї
00000030: 83 C6 10 49 74 19 38 2C 74 F6 A0 B5 07 B4 07 8B | Г╞.It.8,tЎа╡.┤.Л
00000040: F0 AC 3C 00 74 FC BB 07 00 B4 0E CD 10 EB F2 88 | Ём<.t№╗..┤.═.ыЄИ
00000050: 4E 10 E8 46 00 73 2A FE 46 10 80 7E 04 0B 74 0B | N.шF.s*■F.А~..t.

00000060: 80 7E 04 0C 74 05 A0 B6 07 75 D2 80 46 02 06 83 | А~..t.а╢.u╥АF..Г
00000070: 46 08 06 83 56 0A 00 E8 21 00 73 05 A0 B6 07 EB | F..ГV..ш!.s.а╢.ы
00000080: BC 81 3E FE 7D 55 AA 74 0B 80 7E 10 00 74 C8 A0 | ╝Б>■}Uкt.А~..t╚а
00000090: B7 07 EB A9 8B FC 1E 57 8B F5 CB BF 05 00 8A 56 | ╖.ыйЛ№.WЛї╦┐..КV

000000A0: 00 B4 08 CD 13 72 23 8A C1 24 3F 98 8A DE 8A FC | .┤.═.r#К┴$?ШК▐К№

000000B0: 43 F7 E3 8B D1 86 D6 B1 06 D2 EE 42 F7 E2 39 56 | CўуЛ╤Ж╓▒.╥юBўт9V

000000C0: 0A 77 23 72 05 39 46 08 73 1C B8 01 02 BB 00 7C | .w#r.9F.s.╕..╗.|

000000D0: 8B 4E 02 8B 56 00 CD 13 73 51 4F 74 4E 32 E4 8A | ЛN.ЛV.═.sQOtN2фК
000000E0: 56 00 CD 13 EB E4 8A 56 00 60 BB AA 55 B4 41 CD | V.═.ыфКV.`╗кU┤A═

000000F0: 13 72 36 81 FB 55 AA 75 30 F6 C1 01 74 2B 61 60 | .r6Б√Uкu0Ў┴.t+a`

00000100: 6A 00 6A 00 FF 76 0A FF 76 08 6A 00 68 00 7C 6A | j.j. v. v.j.h.|j

00000110: 01 6A 10 B4 42 8B F4 CD 13 61 61 73 0E 4F 74 0B | .j.┤BЛЇ═.aas.Ot.

00000120: 32 E4 8A 56 00 CD 13 EB D6 61 F9 C3 49 6E 76 61 | 2фКV.═.ы╓a∙├Inva

00000130: 6C 69 64 20 70 61 72 74 69 74 69 6F 6E 20 74 61 | lid partition ta

00000140: 62 6C 65 00 45 72 72 6F 72 20 6C 6F 61 64 69 6E | ble.Error loadin

00000150: 67 20 6F 70 65 72 61 74 69 6E 67 20 73 79 73 74 | g operating syst

00000160: 65 6D 00 4D 69 73 73 69 6E 67 20 6F 70 65 72 61 | em.Missing opera

00000170: 74 69 6E 67 20 73 79 73 74 65 6D 00 00 00 00 00 | ting system.....

00000180: 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 | ................

00000190: 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 | ................

000001A0: 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 00 | ................

000001B0: 00 00 00 00 00 2C 44 63 DF 95 0E DB 00 00 80 01 | .....,Dc▀Х.█..А.

000001C0: 01 00 07 FE FF FF 3F 00 00 00 B1 62 A9 03 00 00 | ...■  ?...▒bй...

000001D0: C1 FF 0C FE FF FF F0 62 A9 03 C5 FA 3F 00 00 00 | ┴ .■  Ёbй.┼·?...

000001E0: C1 FF 0F FE FF FF B5 5D E9 03 BE FF 3E 03 00 00 | ┴ .■  ╡]щ.╛ >...

000001F0: 00 00 00 00 00 00 00 00 00 00 00 00 00 00 55 AA | ..............Uк
Пример загрузочного сектора MS-DOS

Ниже представлен шестнадцатеричный дамп загрузочного сектора системной дискеты MS-DOS. Дамп поделён на несколько секций:

1. команда перехода;

2. тип ОС

3. блок параметров BIOS;

4. расширенный блок параметров BIOS;

5. загрузочный код;

6. маркер конца сектора.

00000000: EB 3C 90                                        | ы<Р
00000003: 4D 53 44 4F 53 35 2E 30                         | MSDOS5.0

0000000B: 00 02 01 01 00 02 E0 00 40 0B F0 09 00 12 00 02 | ......р.@.Ё.....

0000001B: 00 00 00 00 00 00 00 00 00                      | .........

00000024: 00 00 29 4C A1 91 22 55 4E 49 4E 53 54 41 4C 4C | ..)LбС"UNINSTALL

00000034: 20 31 46 41 54 31 32 20 20 20                   |  1FAT12   

0000003E: FA 33 C0 8E D0 BC 00 7C 16 07 BB 78 00 36 C5 37 | ·3└О╨╝.|..╗x.6┼7

0000004E: 1E 56 16 53 BF 3E 7C B9 0B 00 FC F3 A4 06 1F C6 | .V.S┐>|╣..№єд..╞

0000005E: 45 FE 0F 8B 0E 18 7C 88 4D F9 89 47 02 C7 07 3E | E■.Л..|ИM∙ЙG.╟.>

0000006E: 7C FB CD 13 72 79 33 C0 39 06 13 7C 74 08 8B 0E | |√═.ry3└9..|t.Л.

0000007E: 13 7C 89 0E 20 7C A0 10 7C F7 26 16 7C 03 06 1C | .|Й. |а.|ў&.|...

0000008E: 7C 13 16 1E 7C 03 06 0E 7C 83 D2 00 A3 50 7C 89 | |...|...|Г╥.гP|Й
0000009E: 16 52 7C A3 49 7C 89 16 4B 7C B8 20 00 F7 26 11 | .R|гI|Й.K|╕ .ў&.

000000AE: 7C 8B 1E 0B 7C 03 C3 48 F7 F3 01 06 49 7C 83 16 | |Л..|.├Hўє..I|Г.

000000BE: 4B 7C 00 BB 00 05 8B 16 52 7C A1 50 7C E8 92 00 | K|.╗..Л.R|бP|шТ.

000000CE: 72 1D B0 01 E8 AC 00 72 16 8B FB B9 0B 00 BE E6 | r.░.шм.r.Л√╣..╛ц
000000DE: 7D F3 A6 75 0A 8D 7F 20 B9 0B 00 F3 A6 74 18 BE | }єжu.Н� ╣..єжt.╛

000000EE: 9E 7D E8 5F 00 33 C0 CD 16 5E 1F 8F 04 8F 44 02 | Ю}ш_.3└═.^.П.ПD.

000000FE: CD 19 58 58 58 EB E8 8B 47 1A 48 48 8A 1E 0D 7C | ═.XXXышЛG.HHК..|

0000010E: 32 FF F7 E3 03 06 49 7C 13 16 4B 7C BB 00 07 B9 | 2 ўу..I|..K|╗..╣

0000011E: 03 00 50 52 51 E8 3A 00 72 D8 B0 01 E8 54 00 59 | ..PRQш:.r╪░.шT.Y

0000012E: 5A 58 72 BB 05 01 00 83 D2 00 03 1E 0B 7C E2 E2 | ZXr╗...Г╥....|тт
0000013E: 8A 2E 15 7C 8A 16 24 7C 8B 1E 49 7C A1 4B 7C EA | К..|К.$|Л.I|бK|ъ
0000014E: 00 00 70 00 AC 0A C0 74 29 B4 0E BB 07 00 CD 10 | ..p.м.└t)┤.╗..═.

0000015E: EB F2 3B 16 18 7C 73 19 F7 36 18 7C FE C2 88 16 | ыЄ;..|s.ў6.|■┬И.

0000016E: 4F 7C 33 D2 F7 36 1A 7C 88 16 25 7C A3 4D 7C F8 | O|3╥ў6.|И.%|гM|°

0000017E: C3 F9 C3 B4 02 8B 16 4D 7C B1 06 D2 E6 0A 36 4F | ├∙├┤.Л.M|▒.╥ц.6O

0000018E: 7C 8B CA 86 E9 8A 16 24 7C 8A 36 25 7C CD 13 C3 | |Л╩ЖщК.$|К6%|═.├

0000019E: 0D 0A 4E 6F 6E 2D 53 79 73 74 65 6D 20 64 69 73 | ..Non-System dis

000001AE: 6B 20 6F 72 20 64 69 73 6B 20 65 72 72 6F 72 0D | k or disk error.

000001BE: 0A 52 65 70 6C 61 63 65 20 61 6E 64 20 70 72 65 | .Replace and pre

000001CE: 73 73 20 61 6E 79 20 6B 65 79 20 77 68 65 6E 20 | ss any key when 

000001DE: 72 65 61 64 79 0D 0A 00 49 4F 20 20 20 20 20 20 | ready...IO      

000001EE: 53 59 53 4D 53 44 4F 53 20 20 20 53 59 53 00 00 | SYSMSDOS   SYS..

000001FE: 55 AA                                           | Uк

Далее приведён исходный текст на ассемблере загрузочного сектора с комментариями.

CS:7C00   EB3C         jmp    7C3E

; переход на начало программы

CS:7C02   90           nop

-------- OEM Name and Version --------

CS:7C03   db 'MSDOS5.0'

-------- BPB и EBPB --------

CS:7C0B   dw 0200h

; Bytes Per Sector

CS:7C0D   db 01h


; Sectors Per Cluster

CS:7C0E   dw 0001h

; Reserved Sectors

CS:7C10   db 02h


; Number of FATs

CS:7C11   dw 00E0h

; Root Entries

CS:7C13   dw 0B40h

; Small Sectors

CS:7C15   db 0F0h


; Media Type

CS:7C16   dw 0009h

; Sectors Per FAT

CS:7C18   dw 0012h

; Sectors Per Track

CS:7C1A   dw 0002h

; Number of Heads

CS:7C1C   dd 00000000h

; Hidden Sectors

CS:7C20   dd 00000000h

; Large Sectors

CS:7C24   db 00h


; Physical Disk Number

CS:7C25   db 00h


; Current Head

CS:7C26   db 29h


; Signature

CS:7C27   dd 2291A14Ch

; Volume Serial Number

CS:7C2B   db 'NO NAME    '
; Volume Label

CS:7C36   db 'FAT12   '

; System ID

-------- Начало программы загрузки --------

; запрет маскируемых прерываний

CS:7C3E   FA           cli




CS:7C3F   33C0         xor    ax, ax
; настройка стекового

CS:7C41   8ED0         mov    ss, ax
; сегмента на начало памяти

CS:7C43   BC007C       mov    sp, 7C00
; установка вершины стека

CS:7C46   16           push   ss

; настройка ES

CS:7C47   07           pop    es

; на начало памяти

CS:7C48   BB7800       mov    bx, 0078
; в BX - смещение вектора 1Eh

; в DS:SI - адрес таблицы параметров дискет 

CS:7C4B   36C537       lds    si, ss:[bx]


CS:7C4E   1E           push   ds

; сохранение DS

CS:7C4F   56           push   si

; сохранение SI

CS:7C50   16           push   ss

; сохранение SS

CS:7C51   53           push   bx

; сохранение BX

; в DI - смещение начала программы (для хранения 
; промежуточных значений)

CS:7C52   BF3E7C       mov    di, 7C3E


; в CX - размер таблицы параметров дискет


CS:7C55   B90B00       mov    cx, 000B


CS:7C58   FC           cld


; сброс флага направления

; копирование таблицы параметров дискет в начало программы 

; (оно затирается)

CS:7C59   F3A4         rep movsb



CS:7C5B   06           push   es

; настройка DS

CS:7C5C   1F           pop    ds

; на начало памяти

; запись в таблицу параметров дискет времени установки головки 15 мс.

CS:7C5D   C645FE0F     mov    byte ptr [di-02], 0F       

; запись в CX числа секторов на дорожке 

CS:7C61   8B0E187C     mov    cx, [7C18]


; запись в таблицу параметров дискет номера последнего сектора 
; на дорожке 

CS:7C65   884DF9       mov    [di-07], cl
; установка нового значения адреса 

CS:7C68   894702       mov    [bx+02], ax


CS:7C6B   C7073E7C     mov    [bx], 7C3E
; таблицы параметров дискет

; разрешение маскируемых прерываний 

CS:7C6F   FB           sti




; INT 13h, функция 00h - сброс контроллера дисководов 

CS:7C70   CD13         int    13



CS:7C72   7279         jb     7CED

; переход по ошибке

CS:7C74   33C0         xor    ax, ax
; обнуление AX

; проверка, используется ли поле BPB.LargeSectors 

CS:7C76   3906137C     cmp    [7C13], ax


CS:7C7A   7408         je     7C84

; переход, если используется

; в CX - число секторов на дискете 

CS:7C7C   8B0E137C     mov    cx, [7C13]


; запись в BPB.LargeSectors значения из BPB.SmallSectors 

CS:7C80   890E207C     mov    [7C20], cx


CS:7C84   A0107C       mov    al, [7C10]
; запись в AL количества FAT

; умножение на количество секторов в FAT 

CS:7C87   F726167C     mul    word ptr [7C16]


CS:7C8B   03061C7C     add    ax, [7C1C]
; прибавление к DX:AX 

CS:7C8F   13161E7C     adc    dx, [7C1E]
; поля BPB.HiddenSectors

CS:7C93   03060E7C     add    ax, [7C0E]
; прибавление к DX:AX

; поля BPB.ReservedSectors

CS:7C97   83D200       adc    dx, 0000


CS:7C9A   A3507C       mov    [7C50], ax
; сохранение DX:AX

CS:7C9D   8916527C     mov    [7C52], dx
; во временных ячейках памяти

CS:7CA1   A3497C       mov    [7C49], ax
; сохранение DX:AX

CS:7CA4   89164B7C     mov    [7C4B], dx
; во временных ячейках памяти

; в AX - размер элемента каталога (32 байта)

CS:7CA8   B82000       mov    ax, 0020



; умножение на количество элементов в DIR 

CS:7CAB   F726117C     mul    word ptr [7C11]


; запись в BX количества байт в секторе 

CS:7CAF   8B1E0B7C     mov    bx, [7C0B]


; добавление к AX количества байт в секторе 

CS:7CB3   03C3         add    ax, bx



CS:7CB5   48           dec    ax

; AX := AX – 1

; вычисление количества секторов, занимаемых DIR 

CS:7CB6   F7F3         div    bx



; добавление к номеру начального сектора DIR

CS:7CB8   0106497C     add    [7C49], ax

; числа секторов в DIR 
; (для получения номера начального сектора IO.SYS)

CS:7CBC   83164B7C00   adc    [7C4B], 0000


; в BX - адрес буфера для считывания первого сектора DIR 
; (cразу после BIOS Data Area)

CS:7CC1   BB0005       mov    bx, 0500



CS:7CC4   8B16527C     mov    dx, [7C52]
; запись в DX:AX номера 

CS:7CC8   A1507C       mov    ax, [7C50]
; начального сектора DIR

CS:7CCB   E89200       call   7D60

; перевод номера логического сектора DIR в трёхмерную геометрию CHS

CS:7CCE   721D         jb     7CED

; переход по ошибке

; в AL - количество секторов для считывания 

CS:7CD0   B001         mov    al, 01



CS:7CD2   E8AC00       call   7D81

; чтение первого сектора DIR

CS:7CD5   7216         jb     7CED

; переход по ошибке

CS:7CD7   8BFB         mov    di, bx
; в DI - смещение буфера

; в CX - 11 (количество байт для сравнения)

CS:7CD9   B90B00       mov    cx, 000B


; в SI - смещение строчки "IO      SYS"

CS:7CDC   BEE67D       mov    si, 7DE6


; сравнение первого элемента DIR с заданной строчкой 

CS:7CDF   F3A6         repz cmpsb



CS:7CE1   750A         jne    7CED

; переход, если не равны

; в DI - смещение второго элемента DIR

CS:7CE3   8D7F20       lea    di, [bx+20] 

; в CX - 11 (количество байт для сравнения)

CS:7CE6   B90B00       mov    cx, 000B


; сравнение второго элемента DIR со строчкой "MSDOS   SYS"


CS:7CE9   F3A6         repz cmpsb



CS:7CEB   7418         je     7D05

; переход, если равны

; в SI - смещение сообщения "Non-System disk or disk error"

CS:7CED   BE9E7D       mov    si, 7D9E



CS:7CF0   E85F00       call   7D52

; вывод строчки на экран

CS:7CF3   33C0         xor    ax, ax
; обнуление AX

; INT 16h, функция 00h - чтение символа с ожиданием нажатия 

CS:7CF5   CD16         int    16



CS:7CF7   5E           pop    si

; восстановление SI

CS:7CF8   1F           pop    ds

; восстановление DS

; восстановление [SI] - там было записано SI 

CS:7CF9   8F04         pop    [si]



; восстановление [SI] - там было записано DS 

CS:7CFB   8F4402       pop    [si+02]


; INT 19h - вызов начального загрузчика


CS:7CFE   CD19         int    19



CS:7D00   58           pop    ax

; восстановление AX

CS:7D01   58           pop    ax

; восстановление AX

CS:7D02   58           pop    ax

; восстановление AX

CS:7D03   EBE8         jmp    7CED

; переход по ошибке

; в AX - номер начального кластера файла IO.SYS 

CS:7D05   8B471A       mov    ax, [bx+1A]


CS:7D08   48           dec    ax

; AX := AX - 1

CS:7D09   48           dec    ax

; AX := AX – 1

; в BL - количество секторов на дорожке 

CS:7D0A   8A1E0D7C     mov    bl, [7C0D]


CS:7D0E   32FF         xor    bh, bh
; обнуление BH

; умножение номера кластера на количество секторов на дорожке

CS:7D10   F7E3         mul    bx



CS:7D12   0306497C     add    ax, [7C49]
; добавление к DX:AX

; номера начального сектора DIR 

CS:7D16   13164B7C     adc    dx, [7C4B]


; в BX - адрес буфера для IO.SYS (сразу после буфера для DIR)

CS:7D1A   BB0007       mov    bx, 0700



; в CX - количество секторов для чтения 

CS:7D1D   B90300       mov    cx, 0003



CS:7D20   50           push   ax

; сохранение AX

CS:7D21   52           push   dx

; сохранение DX

CS:7D22   51           push   cx

; сохранение CX

; перевод номера логического сектора в трёхмерную геометрию CHS 

CS:7D23   E83A00       call   7D60



; переход по ошибке с восстановлением регистров 

CS:7D26   72D8         jb     7D00



; количество секторов для считывания

CS:7D28   B001         mov    al, 01


; считывание очередного сектора IO.SYS


CS:7D2A   E85400       call   7D81



CS:7D2D   59           pop    cx

; восстановление CX

CS:7D2E   5A           pop    dx

; восстановление DX

CS:7D2F   58           pop    ax

; восстановление AX

CS:7D30   72BB         jb     7CED

; переход по ошибке

; увеличение логического 

CS:7D32   050100       add    ax, 0001



CS:7D35   83D200       adc    dx, 0000
; номера сектора

; увеличение указателя на буфер (BX) на размер сектора 

CS:7D38   031E0B7C     add    bx, [7C0B] 

CS:7D3C   E2E2         loop   7D20      
; цикл по секторам

CS:7D3E   8A2E157C     mov    ch, [7C15]
; в CH - тип носителя

CS:7D42   8A16247C     mov    dl, [7C24]
; в DL - номер дисковода

CS:7D46   8B1E497C     mov    bx, [7C49]
; в BX:AX - начальный 

CS:7D4A   A14B7C       mov    ax, [7C4B]
; начальный сектор IO.SYS

; передача управления IO.SYS 

CS:7D4D   EA00007000   jmp    0070:0000



-------- Процедура вывода сообщения об ошибке --------

; загрузка очередного байта сообщения 

CS:7D52   AC           lodsb




; если он равен нулю,

CS:7D53   0AC0         or     al, al


; переход на возврат из процедуры


CS:7D55   7429         je     7D80



CS:7D57   B40E         mov    ah, 0E
; в AH - номер функции 0Eh

; в BX - атрибут выводимого символа 

CS:7D59   BB0700       mov    bx, 0007


; INT 10h, функция 0Eh - вывод символа в режиме телетайпа 

CS:7D5C   CD10         int    10



CS:7D5E   EBF2         jmp    7D52

; цикл по строке сообщения

-------- Процедура перевода логического номера сектора в трёхмерную геометрию CHS --------

; сравнение DX с количеством секторов на дорожке 

CS:7D60   3B16187C     cmp    dx, [7C18]


; переход, если DX >= количество секторов на дорожке 

CS:7D64   7319         jnb    7D7F



; деление DX:AX на количество секторов на дорожке 

CS:7D66   F736187C     div    word ptr [7C18]

; DL := DL + 1, т. к. номера физических секторов 
; начинаются с 1, а не с 0

CS:7D6A   FEC2         inc    dl



; запись номера сектора в ячейку памяти

CS:7D6C   88164F7C     mov    [7C4F], dl


CS:7D70   33D2         xor    dx, dx
; обнуление DX

; деление номера дорожки на количество головок 

CS:7D72   F7361A7C     div    [7C1A]


; запись номера головки в ячейку памяти


CS:7D76   8816257C     mov    [7C25], dl


; запись номера дорожки в ячейку памяти 

CS:7D7A   A34D7C       mov    [7C4D], ax


; сброс флага переноса (нет ошибки)

CS:7D7D   F8           clc




CS:7D7E   C3           ret


; возврат из процедуры

CS:7D7F   F9           stc

; установка флага переноса (ошибка)

CS:7D80   C3           ret


; возврат из процедуры

-------- Процедура чтения сектора --------

; AH = 02h - функция чтения сектора 

CS:7D81   B402         mov    ah, 02


CS:7D83   8B164D7C     mov    dx, [7C4D]
; в DX - номер дорожки

; в CL - 6 (для сдвига на 6 разрядов)

CS:7D87   B106         mov    cl, 06



; в DH - старшие 2 бита номера дорожки 

CS:7D89   D2E6         shl    dh, cl


; младшие 2 бита DH - номер сектора


CS:7D8B   0A364F7C     or     dh, [7C4F]


; в CX - номер сектора и дорожки 

CS:7D8F   8BCA         mov    cx, dx



; в CL - номер сектора и старшие 2 бита номера дорожки, 
; в CH - младшие 8 бит номера дорожки 

CS:7D91   86E9         xchg   cl, ch



CS:7D93   8A16247C     mov    dl, [7C24]
; в DL - номер дисковода

; в DH - номер текущей головки 

CS:7D97   8A36257C     mov    dh, [7C25]


; INT 13h, функция 02h - чтение секторов 

CS:7D9B   CD13         int    13



CS:7D9D   C3           ret


; возврат из процедуры

-------- Текстовые строки --------

CS:7D9E   db 0Dh,0Ah,'Non-System disk or disk error'

CS:7DBF   db 0Dh,0Ah,'Replace and press any key when ready',0Dh,0Ah,00h

CS:7DE6   db 'IO      SYS'

CS:7DF1   db 'MSDOS   SYS'

CS:7DFC   dw 0

-------- Маркер конца сектора --------

CS:7DFE   dw 0AA55h


В общих чертах, алгоритм работы данной программы можно представить следующим образом:

1. Настройка стека, сохранение значений регистров, инициализация программы.

2. Вычисление номера начального сектора корневого каталога по формуле: DIRSector = HiddenSectors + ReservedSectors + NumberOfFATs * SectorsPerFat.

3. Считывание первого сектора корневого каталога в память.

4. Определение, установлена ли MS-DOS на этом диске. Если нет, то вывод соответствующего сообщения и перезагрузка компьютера по нажатию клавиши.

5. Вычисление номера начального сектора файла IO.SYS. Номер Начальный сектора вычисляется с использованием номера первого кластера IO.SYS (номер кластера берётся из элемента корневого каталога для IO.SYS).

6. Считывание части IO.SYS в память.

7. Передача управления IO.SYS.

Контрольные вопросы

1. Для чего служит загрузочный сектор?

2. Какие данные хранятся в блоках параметров BIOS?

3. Для чего служит главная загрузочная запись?

4. Какая информация содержится в таблице разделов?

5. Поясните, каким образом организуются логические диски в расширенном разделе.

6. Изложите в общих чертах алгоритм работы загрузочного кода MS-DOS.

